

Emergent Stories Based on Autonomous Characters with Mindreading Capabilities

Tibor Bosse, Zulfiqar A. Memon, Jan Treur

Vrije Universiteit Amsterdam, Department of Artificial Intelligence
De Boelelaan 1081a, 1081 HV Amsterdam, The Netherlands
<http://www.few.vu.nl/~{tbosse, zamemon, treur}>

Abstract. Virtual storytelling often uses a fixed, pre-scripted storyline, constraining the characters' autonomy and behaviour. An agent-based approach provides possibilities of a narrative emerging from interaction between a number of autonomous characters. In this paper such an approach is proposed, based on autonomous virtual agents that have mindreading capabilities. The approach is applied to generate out of a number of interacting characters a soap storyline, in which some of the characters use mindreading in order to mislead and manipulate the other characters.

1 Introduction

Virtual storytelling addresses automated generation of stories with characters showing more or less human-like behaviour. Both researchers and developers (e.g., games designers or screen writers) work on this topic. Examples of approaches to develop virtual stories are [8], [9], [20], [21], [22]. Traditionally these approaches involve stories with a fixed, pre-scripted storyline constraining the characters in their autonomy and behaviour, but recently more interest is shown in emergent narrative, i.e., storylines generated by interaction between a number of characters with certain personalities that (inter)act as autonomous agents [1]. In this approach, ideally, the designer only determines which (types of) characters will occur in the play and what world events are offered to them.

When a story is generated by autonomous characters, this imposes higher requirements on their capabilities than in case of a prescribed storyline. They should be able to behave in more human-like manners to get a realistic storyline, and need more complex personalities with human-like properties such as emotions and theories of mind, e.g., [12], [13]. To accomplish this, more sophisticated computational cognitive models are to be incorporated within such virtual characters, e.g., [7], [16], [17], [18]. This is a challenging area to be further explored.

This paper explores the possibilities to equip the characters involved in virtual stories in particular with mindreading capabilities; see e.g., [2], [10], [11], [12], [15]. By offering virtual agents such capabilities, they will be able to select behaviours that are useful in social func-

tioning in a more human-like manner, with more freedom of choice, thus enhancing the emergent narrative effect. One of the ways to model an agent B exploiting mindreading of an agent A is to use beliefs, desires and intentions (BDI) as concepts to describe agent A's cognitive processes and actions within agent B. To model the agent B's own behaviour a BDI-model can be used as well; see also [5].

This type of model is exploited in this paper to obtain characters acting in an emergent soap story. The example soap story addressed concentrates on four characters: Boy A, Boy B, Girl A, Girl B. As Boy B is attracted to Girl A, who however is dating Boy A, he exploits mindreading capabilities to come to more favourable circumstances, involving Girl B as well.

As a vehicle the modelling language LEADSTO [4] is used; this language is based on temporal dependencies between two state properties in successive states modelled by *executable dynamic properties*. Let α and β be state properties of the form 'conjunction of ground atoms or negations of ground atoms', then the notation $\alpha \rightarrow_{e, f, g, h} \beta$, means:

If state property α holds for a time interval with duration g , then after some delay (between e and f) state property β will hold for a time interval of length h .

Atomic state properties can have a qualitative format, such as an expression *desire(d)*, expressing that desire d occurs, or a quantitative format such as an expression *has_value(x, v)* expressing that the variable with name x has (real or integer) number v as value.

In this paper, first in Section 2 the general BDI-model is explained, and three specific models for characters are specified based on the BDI-model (Boy A, Girl A, Girl B). Section 3 describes how the BDI-model can also be used to define a model for a mindreading character, resulting in the fourth character model, for Boy B. In Section 4 it is discussed how based on the different models of autonomous characters, multi-agent simulation experiments have been conducted, resulting in an emergent storyline. Section 5 addresses formal analysis of the underlying multi-agent system. Section 6 is a discussion.

2 BDI-Based Character Models

The generic BDI-model which bases the preparation and performing of actions on beliefs, desires and intentions (e.g., [19]) is briefly discussed, after which it is shown how it can be applied to model autonomous characters and their behaviour.

2.1 The Generic BDI-model Used

Figure 1 depicts the generic structure of the BDI-model in causal-graph-like style. Here the box indicates the borders of the agent, the circles denote state properties, and the arrows indicate dynamic properties expressing that one state property leads to (or causes) another state property. In this model, an action A is performed when the subject has the intention to do this action and it has a belief B2 that certain circumstances in the world are fulfilled such that there is an opportunity to do the action. Beliefs are created on the basis of input in the form of observations or communicated information.




Figure 1: Structure of the generic BDI-model.

The intention to do a specific type of action A is generated if there is some desire D, and there is a belief B1 that there are certain circumstances in the world state, that suggest that performing the action A will fulfil desire D (a kind of rationality criterion).

Relations in the BDI-model are specified in informal and formal LEADSTO format as follows:

If D is desired and B1 believed,
then the intention for action A will occur
If action A is intended and B2 believed,
then the action A will be performed

$$\begin{aligned} \text{desire}(D) \wedge \text{belief}(B1) &\rightarrow \text{intention}(P) \\ \text{intention}(P) \wedge \text{belief}(B2) &\rightarrow \text{performs}(P) \end{aligned}$$

Here \wedge stands for the conjunction (and) between the atomic state properties (in the graphical format denoted by an arc connecting two (or more) arrows). As a generic template, including a reference to the agent x concerned, this can be expressed as follows. For any desire D, world state property Z, and action Y such that $\text{has_reason_for}(X, D, Z, Y)$ holds:

$$\text{desire}(X, D) \wedge \text{belief}(X, Z) \rightarrow \text{intention}(X, Y)$$

For any world state property Z and action Y such that $\text{is_opportunity_for}(X, Z, Y)$ holds:

$$\text{intention}(X, Y) \wedge \text{belief}(X, Z) \rightarrow \text{performs}(X, Y)$$

Here $\text{has_reason_for}(X, D, Z, Y)$ is a relation that can be used to specify which state property Z is considered a reason to choose a certain intention Y for desire D. Similarly $\text{is_opportunity_for}(X, Z, Y)$ is a relation that can be used to specify which state property Z is considered an opportunity to actually perform an intended action Y.

2.2 BDI-Based Characters and their Behaviour

The BDI-model was used to model autonomous characters. The first character shown as an illustration is Boy A. This is a character that tries to date any attractive girl, as defined by the following BDI-model.

BABD1 Boy A Belief Determination

If Boy A observes some world fact then he will believe this world fact
If Boy A gets some world fact communicated, then he will believe this world fact

$$\begin{aligned} \text{observed_result}(\text{BoyA}, I) &\rightarrow \text{belief}(\text{BoyA}, I) \\ \text{communicated_from_to}(I, Y, \text{BoyA}) &\rightarrow \text{belief}(\text{BoyA}, I) \end{aligned}$$

BADD1 Boy A Desire Determination

If Boy A believes that a girl is attractive, then he will desire to have contact with her

$$\text{belief}(\text{BoyA}, \text{is_attractive}(Y)) \rightarrow \text{desire}(\text{BoyA}, \text{contact_with}(Y))$$

BAID1 Boy A Intention Determination

If Boy A desires to have contact with a girl, and he believes that she shows interest, then he will have the intention to date her

$$\begin{aligned} \text{desire}(\text{BoyA}, \text{contact_with}(Y)) \wedge \text{belief}(\text{BoyA}, \text{shows_interest}(Y)) \\ \rightarrow \text{intention}(\text{BoyA}, \text{date}(Y)) \end{aligned}$$

BAAD1 Boy A Action Determination

If Boy A intends to date a girl, and he believes that she suggests to date, then he will date her

$$\begin{aligned} \text{intention}(\text{BoyA}, \text{date}(Y)) \wedge \text{belief}(\text{BoyA}, \text{suggest_to_date}(Y)) \\ \rightarrow \text{performs}(\text{BoyA}, \text{date}(Y)) \end{aligned}$$

Using the visualisation template provided in Figure 1, this model can be depicted as shown in Figure 2.




Figure 2: BDI-model for Boy A.

Note that this BDI-model defines an internal view on the agent's cognitive processes. The external behaviour view generated by these processes is defined by:

BA1 Boy A Behaviour External View

If Boy A observes an attractive girl that shows interest in him and suggests to date, then he will date her.

```
observed_result(BoyA, is_attractive(Y) ^ shows_interest(Y)) ^  
communicated_from_to(suggest_to_date, Y, BoyA)  
→ performs(BoyA, date(Y))
```

Often characters for virtual stories or plays are defined by an external behavioural view only. However, an internal view provides the possibility in the story to provide, in addition to the character's behaviour, more insight in the internal world of the character (e.g., the motivations behind the behaviour). This is one of the reasons to choose for a BDI-based internal model for the characters. The second character, girl A, is looking for a partner of a good family, who does not date other girls and is rich (a good job), as expressed in the following BDI-model. Formalisations are sometimes left out due to space limitations.

GABD1 Girl A Belief Determination

If Girl A observes some world fact then she will believe this world fact
If Girl A gets some world fact communicated, then she will believe this world fact

GADD1 Girl A Desire Determination

If Girl A believes that a certain boy has a good family and does not date other girls, then she will desire to be with him

GAID1 Girl A Intention Determination

If Girl A desires to be with a boy, and he is rich, then she will have the intention to date him

GAAD1 Girl A Action Determination

If Girl A intends to date a boy, and he proposes to date, then she will date him

The external behaviour view generated by these is:

GA1 Girl A Behaviour External View

If Girl A observes a boy that is from a good family, rich, does not date other girls, and he suggests to date, then she will date him.

```
observed_result(GirlA, good_family(X) ^ is_rich(X) ^ not_dating(X)) ^  
communicated_from_to(suggest_to_date, X, GirlA)  
→ performs(GirlA, date(X))
```

The third character discussed is a type of poor girl who has no good family: Girl B, defined as follows.

GBBD1 Girl B Belief Determination

If Girl B observes some world fact then she will believe this world fact
If Girl A gets some world fact communicated, then she will believe this world fact

GBDD1 Girl B Desire Determination

If Girl B believes that she has no good family and no good job, then she will desire to get money.

GBID1 Girl B Intention Determination

If Girl B desires to get money, and she believes somebody offers her money to date a boy, then she will have the intention to show interest in that boy and suggest to date

GBAD1 Girl B Action Determination

If Girl B intends to date a boy, and he is around, then she will show interest in him and suggest him to date

The external behaviour view generated by these is:

GB1 Girl B Behaviour External View

If Girl B is offered money to date any boy, then she will show interest in this boy and suggest him to date.

```
observed_result(no_good_family ^ no_good_job ^ BoyA_is_around) ^  
communicated_from_to(money_offer_for_dating(X2), X1, GirlB)  
→ performs(GirlB, show_interest_in(X2)) ^  
communication_from_to(suggest_to_date, GirlB, X2)
```

The fourth character used in the example story presented, Boy B, also has been specified by a BDI-model. However, this BDI-model is more complex, as it uses mindreading capabilities in the desire determination process, and is discussed in next section.

3 The Model of a Mindreading Character

Mindreading or Theory of Mind [2], [10], [11], [12], [15] is the capability of an agent to attribute mental states such as beliefs, desires and intentions to other agents. Thus, for an agent B a Theory of Mind may use concepts for agent A's beliefs, desires and intentions. Moreover, the model may incorporate BDI-concepts for agent B's own beliefs, desires and intentions as well. By doing this, agent B is able to not only have a theory about the mind of agent A, but also to use it within its own BDI-based cognitive processes to generate its actions, for example, to manipulate the other agent.

3.1 Representing a Theory of Mind within an Agent

To represent a Theory of Mind on agent A within an agent B, a number of nested expressions are used, e.g.:

```
belief(B, desire(A, D))
```

This expresses that agent B believes that agent A has desire D. Moreover,

```
desire(B, not(intention(A, X)))
```

This expresses that agent B desires that agent A does not intend action X.

```
belief(B, leads_to(con(intention(A, P), belief(A, B2)), performs(A, P), D))
```

This expresses (where *con* indicates a conjunction):

Agent B believes that, when A has the intention to perform action P and the belief that B2 holds, then within time duration D (which is assumed small) it will perform action P.

This represents a statement about a *dynamic property*, rather than a statement about a state property. In this manner temporal relations depicted in a graph as in Figure 1 can be represented. Character Boy B's Theory of Mind for Girl A, resp. Girl B, Boy A is expressed by:

BBToMGA Boy B Theory of Mind for Girl A

```
belief(BoyB, leads_to(observed_result(GirlA, X), belief(GirlA, X), D))  
belief(BoyB, leads_to(communicated_from_to(X, Y, GirlA),  
belief(GirlA, X), D))  
belief(BoyB, leads_to(con(belief(GirlA, good_family(Y)),  
belief(GirlA, not_dating(Y))), desire(GirlA, be_with(Y)), D))  
belief(BoyB, leads_to(con(desire(GirlA, be_with(Y)),  
belief(GirlA, rich(Y))), intention(GirlA, date(Y)), D))  
belief(BoyB, leads_to(con(intention(GirlA, date(Y)),  
belief(GirlA, suggests_to_date(Y))),  
performs(GirlA, date(Y)), D))
```

BBToMGB Boy B Theory of Mind for Girl B

```
belief(BoyB, leads_to(observed_result(GirlB, X), belief(GirlB, X), D))  
belief(BoyB, leads_to(communicated_from_to(X, Y, GirlB),  
belief(GirlB, X), D))  
belief(BoyB, leads_to(con(belief(GirlB, poor_family,  
belief(GirlB, no_good_job))),
```

```

    desire(GirlB, get_money), D))
belief(BoyB, leads_to(con(desire(GirlB, get_money),
    belief(GirlB, money_offered_for_dating(Y))),
    intention(GirlB, show_interest_in_and_suggest_to_date(Y)), D))
belief(BoyB, leads_to(con(intention(GirlB,
    show_interest_in_and_suggest_to_date(Y)),
    belief(GirlB, is_around(Y))),
    performs(GirlB, show_interest_in(Y) ^
    communication_from_to(suggest_to_date, GirlB, Y), D))

```

BBToMGA Boy B Theory of Mind for Boy A

```

belief(BoyB, leads_to(observed_result(BoyA, X), belief(BoyA, X), D))
belief(BoyB, leads_to(communicated_from_to(X, Y, BoyA),
    belief(BoyA, X), D))
belief(BoyB, leads_to(belief(BoyA, is_attractive(Y))),
    desire(BoyA, contact_with(Y), D))
belief(BoyB, leads_to(con(desire(BoyA, contact_with(Y)),
    belief(BoyA, shows_interest(Y))),
    intention(BoyA, date(Y), D))
belief(BoyB, leads_to(con(intention(BoyA, date(Y)),
    belief(BoyA, suggests_to_date(Y))),
    performs(BoyA, date(Y), D))

```

3.2 Generating Desires in a Mindreading Character

When an agent B has a Theory of Mind of another character, it can take this into account in generating its own desires. For example, if agent B desires some action of agent A to take place, as a consequence it can generate a desire that agent A has the intention to perform this action. From the latter desire agent B can generate the desire that agent A has a desire that is fulfilled by the action, and so on. The following generic specifications enable such desire generation processes in agent B's BDI-model (here X and Y may be negations of other statements):

BBToMX

If agent B believes that X leads to Y and desires Y, then it will desire X.

If agent B believes that X leads to Y and it desires not Y, then it will desire not X.

If agent B desires the conjunction of X1 and X2 and it believes that X1 and X2 are not fixed, then it will desire X1 and it will desire X2.

If agent B desires the conjunction of X1 and X2 and it believes that X1 is fixed, then it will desire X2.

If agent B desires the conjunction of X1 and X2 and it believes that X2 is fixed, then it will desire X1.

If agent B believes that X is fixed and that X leads to Y, then it will believe that Y is fixed.

```

desire(B, Y) ^ belief(B, leads_to(X, Y, D)) → desire(B, X)
desire(B, not(Y)) ^ belief(B, leads_to(X, Y, D)) → desire(B, not(X))
desire(B, con(X1, X2)) ^ not belief(B, fixed_true(X1)) ^
    not belief(B, fixed_true(X2)) → desire(B, X1) ^ desire(B, X2)
desire(B, con(X1, X2)) ^ belief(B, fixed_true(X1)) → desire(B, X2)
desire(B, con(X1, X2)) ^ belief(B, fixed_true(X2)) → desire(B, X1)
belief(B, fixed_true(X)) ^ belief(B, leads_to(X, Y, D)) →
    belief(B, fixed_true(Y))

```

Here a belief on fixed_true(X) expresses that the agent considers the indicated state property X true and unchangeable.

3.3 From Mindreading to Intentions and Actions

Within the model of agent B, the mindreading is used to generate specific desires. For intention and action gen-

eration based on these desires, instantiations of the schemes shown in Section 2.1 are included as well. For the example character Boy B:

BBID1 Boy B Intention Determination

If Boy B desires that girl B gets money for dating Boy A, and he believes he has money, then he will have the intention to offer Girl B money to date Boy A.

BBAD1 Boy B Action Determination

If Boy B intends to offer Girl B money to date Boy A, and he believes she is around, then he will offer her money to date Boy A.

Moreover, as before Boy B has specifications for beliefs.

BBBD1 Boy B Belief Determination

If Boy B observes some world fact then he will believe this world fact
 If Boy B gets some world fact communicated, then he will believe this world fact

All of the specifications presented above describe a rather complex internal view on a mindreading agent. In contrast, the external view on character Boy B's behaviour, implied by the internal view, is quite simple: he just has to offer Girl B money to date Boy A.

BB1 Boy B Behaviour External View

At some point in time Boy B will offer money to Girl B to date Boy A.
 $\text{observed_result}(\text{has_money} \wedge \text{GirlB_is_around}) \rightarrow$
 $\text{communication_from_to}(\text{money_offer_for_dating}(\text{BoyA}), \text{BoyB}, \text{GirlB})$

One might be tempted to just ignore the complex internal specification, and define the character Boy B by the simple external view specification instead. For the events in the story this will make no difference. However, as also indicated earlier, leaving out internal models of characters would provide a very inflexible, brittle solution, which is not explainable by referring to the inner world and motivations of the characters.

4 Simulation Results

The BDI-models of the four autonomous characters described in the previous sections have been used to generate emergent virtual storielines. To this end, first the models (represented in LEADSTO) have been used as input for the LEADSTO simulation software environment [4] in order to generate simulation traces (i.e., sequences of events over time). An example of (a part of) such a trace is shown in Figure 3. Here, time is on the horizontal axis, and state properties are on the vertical axis. A dark box indicates that a state property is true. It is shown how Boy A comes to date Girl B.




Figure 3: Example simulation trace

Next, for each of the atomic state properties that occurs in the model, a mapping has been created to a text fragment. For example, the state property

$\text{belief}(X, \text{shows_interest}(Y)),$

corresponds to the text fragment

“X believes that Y shows interest in him”.

Similarly, the state property `desire(X, contact_with(Y))`, corresponds to “X desires to have contact with Y”. Using these mappings and a specific conversion program that has been written, the LEADSTO simulation traces are automatically translated into virtual storielines in textual format; thereby, also the generic names Boy A, et cetera are replaced by story names Al, Ann, Bob and Bo. An example of fragments of such a generated storyline is shown in Table 1. For more details about the simulation and the generation of the storyline, see URL <http://www.few.vu.nl/~treur/storiesappendix.pdf>.

Bob desires to get money.
Al desires to have contact with Bo.
Bob desires that Ann does not date Al.
Bob desires that Ann does not intend to date Al.
Bob desires that Ann does not desire to be with Al.
Bob desires that Ann believes that Al is dating another girl.
Bob desires that Al observes that Al is dating another girl.
Bob desires that Al dates Bo.
Bob desires that Al intends to date Bo.
Bob desires that Al believes that Bo has offered him to date her.
Bob desires that Al believes that Bo shows interest in him.
Bob desires that Al observes that Bo shows interest in him.
Bob desires that Al hears from Bo an offer to date.
Bob desires that Bo shows interest in Al.
Bob desires that Bo suggests Al to date.
Bob desires that Bo intends to suggest Al to date.
Bob desires that Bo believes that he is offering her money to date Al.
Bob desires that Bo observes that he is offering money to date Al.
Bob intends to offer Bo money to date Al.
Bob offers Bo money to date Al.
Bo hears that Bob offers her money to date Al.
Bo believes that Bob offered her money to date Al.
Bo intends to show interest in Al and suggest Al to date.
Bo shows interest in Al.
Bo suggests Al to date.
Al observes that Bo shows interest in him.
Al hears that Bo suggests to date.
Al believes that Bo shows interest in him.
Al believes that Bo suggested to date.
Al intends to date Bo.
Al dates Bo.
Ann observes that Al is dating Bo.
Ann believes that Al is dating Bo.

Table 1: Fragments of a Generated Example Storyline

5 Formal Analysis of Dynamic Properties

When a model for multiple characters such as the one described in Sections 2 and 3, has been specified, it is easy to produce various simulations based on different settings, initial conditions and external events offered. Moreover, it is possible to incorporate nondeterministic behaviours by temporal rules that involve probabilistic

effects (cf. [4]). Thus large sets of traces can be generated. When such a set is given, it is more convenient to check them on interesting (emergent) properties automatically, than going through them by hand. Furthermore, it may also be useful when insight is provided how dynamic properties of the multiagent system as a whole depend on dynamic properties of the agents within the system, and further on, how these relate to properties of specific components within the agents. This section shows how this can be achieved.

In order to analyse whether the resulting stories satisfy interesting (expected or unexpected) properties, a number of *dynamic properties* have been specified for different aggregation levels of the multi-agent system behind the emergent story, cf. [14]. The main property considered for the story as a whole is: will at the end Girl A date Boy B?

GP1 At some point in time Girl A will date Boy B.
 $\text{true} \rightarrow \text{performs}(\text{GirlA}, \text{date}(\text{BoyB}))$

Whether or not this property is fulfilled depends on properties of the agents’ behaviours. Furthermore, these properties of the agents’ behaviours depend on their internal components, in this case components for belief, desire, intention, and action determination. In Figure 4, it is shown how the property GP1 at the highest level relates to properties of the agents, and how properties of the agents relate to properties of their components. The properties are described in more detail below.

The property GP1 of the system as a whole can be logically related to properties of the agents by the following *interlevel relation*:

$$\text{BA1} \ \& \ \text{GA1} \ \& \ \text{BB1} \ \& \ \text{GB1} \ \& \ \text{IP} \Rightarrow \text{GP1}$$

Here IP stands for simple interaction properties expressing that generated output will reach the input of the relevant agent. As indicated in Figure 4, each property of an agent is logically related to properties of the components within the agent. The interlevel relations for resp. Boy A, Girl A, and Girl B are:

$$\begin{aligned} \text{BABD1} \ \& \ \text{BADD1} \ \& \ \text{BAID1} \ \& \ \text{BAAD1} &\Rightarrow \text{BA1} \\ \text{GABD1} \ \& \ \text{GADD1} \ \& \ \text{GAID1} \ \& \ \text{GAAD1} &\Rightarrow \text{GA1} \\ \text{GBBD1} \ \& \ \text{GBDD1} \ \& \ \text{GBID1} \ \& \ \text{GBAD1} &\Rightarrow \text{GB1} \end{aligned}$$

The first interlevel relation, for example, expresses that the behavioural property for Boy A holds when belief, desire, intention, and action determination have the right properties: BABD1, BADD1, BAID1, BAAD1; see Section 2.2. For the other two cases (Girl A, Girl B) it is similar. The interlevel relation within Boy B,

$\text{BBBD1} \ \& \ \text{BBDD1} \ \& \ \text{BBID1} \ \& \ \text{BBAD1} \Rightarrow \text{GA1}$
 involves different elements. The second property is defined as follows (for the other three, see Section 3).




Figure 4: Interlevel relations between dynamic properties in the virtual story.

BBDD1 Boy B Desire Determination

If Boy B desires that Girl A does not date Boy A, then he will desire that she observes that Boy A dates another girl.

If Boy B desires that Girl A observes that Boy A dates another girl, then he will desire that Girl B shows interest in him and suggests him to date.

If Boy B desires that Girl B shows interest in Boy A and suggests him to date, then he will desire that girl B is offered money for dating Boy A.

```

desire(BoyB, not(dates(GirlA, BoyA)))
→ desire(BoyB, observes(GirlA, not(not_dating(BoyA)))
desire(BoyB, observes(GirlA, dates(BoyA, X)))
→ desire(BoyB, observes(BoyA, shows_interest(GirlB)) ∧
communicated_from_to(suggest_to_date, GirlB, BoyA)))
desire(BoyB, observes(BoyA, shows_interest(GirlB)) ∧
communicated_from_to(suggest_to_date, GirlB, BoyA)))
→ desire(BoyB, communicated_from_to(
money_offer_for_dating(BoyA, BoyB, GirlB))

```

This property BBDD1 has another interlevel relation to the properties BBToMBA, BBToMGA, BBToMGB, BBToMX defined in Section 3:

BBToMBA & BBToMGA &
BBToMGB & BBToMX \Rightarrow BBDD1

Using a dedicated Checker Tool (cf. [3]), the dynamic properties as specified above have been automatically verified against the generated storyline shown in Section 4. In this case, they all turned out to hold. However, there may be certain situations in which expected properties (such as “At some point in time, Boy B will date Girl A”) do not hold. In such situations, the Checker Tool and the interlevel relationships between dynamic properties may be used for *diagnosis* of the story. For example, suppose for a given story at some point in time it has been detected (using the Checker Tool) that the dynamic property GP1 does not hold, i.e., Boy B does never date Girl A. Given the AND-tree structure in Figure 4, at least one of the children nodes of GP1 will not hold, which means that either BA1, GA1, BB1, or GB1 will not hold. Suppose by further checking it is found that GB1 does not hold. Then the diagnostic process can be continued by

focusing on this property. It follows that either GBBD1, GBDD1, GBID1, or GBAD1 does not hold. Checking these three properties will pinpoint the cause of the error. Notice that this diagnostic process is economic in the sense that the whole subtree under, e.g., BB1 is not examined since there is no reason for that, as BB1 holds.

6 Discussion

Within virtual storytelling, the idea of emergent narrative is attracting more and more interest: stories in which only a number of autonomous characters and their personalities are given, rather than the precise script of the story [1]. This goes hand in hand with a growing trend to incorporate more sophisticated cognitive models within the characters involved in virtual stories (e.g., [17], [18]). As another step in that direction, the current paper presents an approach to build in mindreading capabilities into virtual characters.

To illustrate the approach, BDI-models have been created for four example characters for a soap story. One of these characters was equipped with more sophisticated mindreading capabilities. The models were specified in LEADSTO format, and have been used as input for the LEADSTO simulation software environment [4] in order to generate simulation traces. For each of the atomic state properties that occurs in the BDI-models, a mapping has been created to a text fragment. Using this mapping and a specific conversion program, the LEADSTO simulation traces have been translated into textual descriptions of (virtual) storylines. Finally, a formal analysis has been performed to find out whether the generated storylines adhere to certain dynamic properties.

In [5] the functioning of humans in organisations was addressed, based on a precursor of the model for Theory of Mind presented here. Besides that a different context was analysed, this precursor model has a number of dif-

ferences from the model here. For example, in that model the agent uses ‘depends on’ relations between mental states of another agent, which is a more rough-grained approach than the approach presented here, based on more precise causal and temporal logical relations between mental states. As a consequence, for example, representations as given in Section 3 are different and much more detailed, and the generation of and focusing on desires is much more specific. A similar difference can be indicated with [6], where also a precursor model based on ‘depends on’ relations was used to analyse mindreading behaviour of animals.

Virtual stories involving autonomous characters with elaborated cognitive or psychological capabilities can be used for a number of purposes. On the one hand, they may be used for entertainment (e.g., for creating computer games with more complex, unpredictable and more human-like characters). On the other hand, they may be used for educational purposes (e.g., to create a virtual training environment for psychotherapists, which enables them to practice anger management sessions with virtual clients). Further research will investigate whether the presented approach is suitable for such purposes. As soon as these types of challenges will be tackled, also a more precise evaluation will be performed of how humans perceive the current characters (e.g. in terms of believability), which was currently ignored.

References

1. Aylett, R. (1999). Narrative in Virtual Environments: Towards Emergent Narrative. In: *Working notes of the Narrative Intelligence Symposium*, AAAI Spring Symposium Series. Menlo Park, California, AAAI Press.
2. Bogdan, R.J. (1997). *Interpreting Minds*. MIT Press.
3. Bosse, T., Jonker, C.M., Meij, L. van der, Sharpanskykh, A., and Treur, J., (2006). Specification and Verification of Dynamics in Cognitive Agent Models. In: *Proc. of the 6th Int. Conf. on Intelligent Agent Technology, IAT'06*. IEEE Computer Society Press, 2006, pp. 247-254.
4. Bosse, T., Jonker, C.M., Meij, L. van der, and Treur, J. (2007). A Language and Environment for Analysis of Dynamics by SimulaTiOn. *Int. Journal of Artificial Intelligence Tools*, vol. 16, 2007, to appear.
5. Bosse, T., Memon, Z.A., and Treur, J., A Two-level BDI-Agent Model for Theory of Mind and its Use in Social Manipulation. In: P. Olivier, C. Kray (eds.), *Proc. of the Artificial and Ambient Intelligence Conference, AISB'07, Mindful Environments Symposium*, 2007, pp. 335-342.
6. Bosse, T., Memon, Z.A., and Treur, J., Modelling Animal Behaviour Based on Interpretation of Another Animal's Behaviour. In: Lewis, R., Polk, T. (eds.), *Proc. of the 8th Intern. Conf. on Cognitive Modeling, ICCM'07*. Taylor and Francis, 2007, to appear.
7. Bosse, T., Pontier, M., Siddiqui, G.F., and Treur, J., Incorporating Emotion Regulation into Virtual Stories. In: Pelachaud, C., et al. (eds.), *Proceedings of the 7th International Conference on Intelligent Virtual Agents*. Lecture Notes in AI, Springer Verlag, 2007, to appear.
8. Cavazza, M., Charles, F., and Mead, S. (2002). Interacting with virtual characters in interactive storytelling. In: *Proceedings of the First International Conference on Autonomous Agents and Multi-agent Systems, AAMAS'02*. ACM Press, pp. 318-325.
9. Dautenhahn, K. (1998). Story-Telling in Virtual Environments. *Proceedings of the ECAI'98 Workshop on Intelligent Virtual Environments*, Brighton, UK.
10. Dennett, D.C. (1987). *The Intentional Stance*. MIT Press. Cambridge Mass.
11. Gärdenfors, P. (2003). *How Homo Became Sapiens: On The Evolution Of Thinking*. Oxford University Press, 2003.
12. Goldman, A.I. (2006). *Simulating Minds: The Philosophy, Psychology and Neuroscience of Mind Reading*. Oxford University Press.
13. Gross, J.J. (ed.) (2007). *Handbook of Emotion Regulation*. New York: Guilford Press.
14. Jonker, C.M., and Treur, J. (2002). Compositional Verification of Multi-Agent Systems: a Formal Analysis of Proactiveness and Reactiveness. *Intern. J. of Cooperative Information Systems*, vol. 11, 2002, pp. 51-92.
15. Malle, B.F., Moses, L.J., Baldwin, D.A. (2001). *Intentions and Intentionality*. MIT Press.
16. Marsella, S. C., Johnson, W. L., and LaBore, C. (2000), Interactive Pedagogical Drama. In: *Proc. of the 4th Intern. Conf. on Autonomous Agents*, ACM Press, pp. 301-308.
17. Marsella, Stacy C., Pynadath, David V., and Read, Stephen J. PsychSim: Agent-based modeling of social interaction and influence. In: *Proc. of the Intern. Conf. on Cognitive Modeling – ICCM 2004*, pp. 243-248.
18. Paiva, A., Machado, I., and Prada, R. (2001). Heroes, villains, magicians, ...: Dramatis personae in a virtual story creation environment. In: *Proc. of the Conference on Intelligent User Interfaces, IUI'01*, pp. 129-136.
19. Rao, A.S. and Georgeff, M.P. (1995) *BDI-agents: from theory to practice*. In *Proceedings of the International Conference on Multagent Systems*, pp. 312 – 319.
20. Robertson, J., and Good, J. (2003). Using a Collaborative Virtual Role-Play Environment to Foster Characterisation in Stories. *J. of Interactive Learning Research*, vol. 14, pp. 5-29.
21. Romano, D.M., Brna, P., Robertson, J., and Louchart, S. (2007). AI and Narrative Games for Education. In: Olivier, P. and Kray, C. (eds.), *Artificial & Ambient Intelligence. Proc. of AISB'07*, pp. 385.
22. Theune, M., Faas, S., Heylen D., and Nijholt, A. (2003). The Virtual Storyteller: Story Creation by Intelligent Agents. *Proc. of Technologies for Interactive Digital Storytelling and Entertainment*, pp. 204-215.